EXAMPLE 4 (part I): pure exchange

The simple two good, two agent model of exchange equilibrium (no production process, only exchange). The world endowments for goods x and y are both equal to 1.



Six parameters (declared as scalars) are used to parameterize the model.
We introduce new utility function CES:

$$U(XA,YA) := \left[THETA\_A \cdot XA^{\frac{SIGMA\_A-1}{SIGMA\_A}} + (1-THETA\_A)\cdot YA^{\frac{SIGMA\_A-1}{SIGMA\_A}}\right]^{\frac{SIGMA\_A}{SIGMA\_A-1}}$$

$$U(XB,YB) := \left[THETA\_B \cdot XB^{\frac{SIGMA\_B-1}{SIGMA\_B}} + (1-THETA\_B)\cdot YB^{\frac{SIGMA\_B-1}{SIGMA\_B}}\right]^{\frac{SIGMA\_B}{SIGMA\_B-1}}$$

$REPORT section requests the solution system to display values for inputs, outputs, final demands or welfare indices at the equilibrium. Example: XAD it is demand by agent A on good X
YAD it is demand by agent A on good Y
where XAD*PX+YAD*PY=A and XBD*PX+YBD*PY=B

```
*       Request the old version of PATH
option mcp=pathold;

*The world endowments for good X and Y are both equal to one.

SCALAR  XA              AGENT A ENDOWMENT OF X ( 0 < XA < 1 )   /0.2/
        YA              AGENT A ENDOWMENT OF Y ( 0 < YA < 1 )   /0.8/
        THETA_A         AGENT A DEMAND SHARE PERAMETER FOR X    /0.5/
        THETA_B         AGENT B DEMAND SHARE PARAMETER FOR X    /0.8/
        SIGMA_A         AGENT A ELASTICITY PARAMETER            /2.0/
        SIGMA_B         AGENT B ELASTICITY PARAMETER            /0.5/;

$ONTEXT

$MODEL:EXCHANGE

$COMMODITIES:
        PX      ! EXCHANGE PRICE OF GOOD X
        PY      ! EXCHANGE PRICE OF GOOD Y

$CONSUMERS:
        A       ! CONSUMER A
        B       ! CONSUMER B

* This model specification uses the default value for reference prices in the
* demand function blocks. When "P:value" is not specified, "P:1" is assumed.

$DEMAND:A       s:SIGMA_A
        E:PX    Q:XA
        E:PY    Q:YA
        D:PX    Q:THETA_A
```

```
        D:PY    Q:(1-THETA_A)
* Any numeric input field in an MPSGE model may be "computed"
*(algebraic expression may be enclosed within parentheses and legitimate GAMS code)

$DEMAND:B       s:SIGMA_B
        E:PX    Q:(1-XA)
        E:PY    Q:(1-YA)
        D:PX    Q:THETA_B
        D:PY    Q:(1-THETA_B)

* The $REPORT section of the input file requests the solution system to return
* values for inputs, outputs, final demands or welfare indices at the equilibrium.
* Only those items which are requested will be written to the solution file.
* Each record in the report block begins with a V: (variable name) field.

$REPORT:
         V:XAD   D:PX    DEMAND:A
         V:YAD   D:PY    DEMAND:A
         V:XBD   D:PX    DEMAND:B
         V:YBD   D:PY    DEMAND:B

$OFFTEXT
$SYSINCLUDE mpsgeset EXCHANGE

$INCLUDE EXCHANGE.GEN
SOLVE EXCHANGE USING MCP;
```

```
                           LOWER       LEVEL       UPPER     MARGINAL

--- VAR PX                   .         1.546       +INF    -6.893E-7
--- VAR PY                   .         0.864       +INF    1.6553E-7
--- VAR A                    .         1.000       +INF    9.2276E-7
--- VAR B                    .         1.409       +INF        .
--- VAR XAD                  .         0.232       +INF        .
--- VAR YAD                  .         0.743       +INF        .
--- VAR XBD                  .         0.768       +INF        .
--- VAR YBD                  .         0.257       +INF        .

 PX  EXCHANGE PRICE OF GOOD X
 PY  EXCHANGE PRICE OF GOOD Y
 A  CONSUMER A
 B  CONSUMER B
```



Conclusion: B has limited elasticity of substitution and comparative advantage in X. A can easy substitute X with Y since he has elasticity of substitution above 1, but he prefer equal share since THETA_A=0.5. Thus XA should increase and YA should decrease. B is satisfied with the initial allocation because THETA_B=0.8, i.e. XB=0.8. If A wants to buy X, he has to offer a high price in order to give the incentive to B to sell this product. At the same time A has to sell Y, but B does not want this product, i.e. A should offer so low price that B will have an incentive to buy Y. Thus PX>PY. Finally income for A=PX*XAD+PY*YAD becomes lower than B=PX*XBD+PY*YBD.

EXAMPLE 4 (part II): second welfare theorem

Absolute levels of income and price are not appropriate for general equilibrium modeling. A CGE model determines only **relative** prices.

```
SCALAR
        PRATIO                 EQUILIBRIUM PRICE X IN TERMS OF Y
        IRATIO                 EQUILIBRIUM RATIO OF CONSUMER A INCOME TO CONSUMER B INCOME;

PRATIO = PX.L / PY.L;
IRATIO = A.L / B.L;

DISPLAY IRATIO, PRATIO;
```

We have to compute an alternative efficient equilibrium where income levels for A and B are equal, to demonstrate that when incomes are both fixed, the equilibrium remains efficient but the connection between market prices and endowment income is eliminated. This will replicate the **Second Welfare Theorem** - any Pareto-efficient allocation can be supported as a price quasi-equilibrium with transfers (under the assumption of convexity for preferences and production set)

```
A.FX = 1;
B.FX = 1;

$INCLUDE EXCHANGE.GEN
SOLVE EXCHANGE USING MCP;

SCALAR
        TRANSFER        IMPLIED TRANSFER FROM A TO B AS A PERCENTAGE OF INCOME;

TRANSFER=100*(A.L- (PX.L*XA +PY.L*YA));
PRATIO = PX.L/PY.L;
IRATIO = A.L/B.L;

DISPLAY TRANSFER, PRATIO, IRATIO;
```

```
                              LOWER     LEVEL     UPPER    MARGINAL

      --- VAR PX                .        1.223     +INF       .
      --- VAR PY                .        0.777     +INF       .
      --- VAR A               1.000      1.000     1.000     0.134
      --- VAR B               1.000      1.000     1.000    -0.134
      --- VAR XAD               .        0.318     +INF       .
      --- VAR YAD               .        0.786     +INF       .
      --- VAR XBD               .        0.682     +INF       .
      --- VAR YBD               .        0.214     +INF       .

       PX  EXCHANGE PRICE OF GOOD X
       PY  EXCHANGE PRICE OF GOOD Y
       A  CONSUMER A
       B  CONSUMER B

      5 PARAMETER TRANSFER            =       13.351  IMPLIED TRANSFER FROM
                                                      A TO B AS A PERCENTAG
                                                      E OF INCOME

        PARAMETER PRATIO              =        1.572  EQUILIBRIUM PRICE X I
                                                      N TERMS OF Y

        PARAMETER IRATIO              =        1.000  EQUILIBRIUM RATIO OF
                                                      CONSUMER A INCOME TO
                                                      CONSUMER B INCOME
```

Details for IRATIO=A/B:

A=PX*XAD+PY*YAD =1,223*0,318+0,777*0,786=1

B= PX*XBD +PY*YBD = 1,223*0,682+0,777*0,214=1



PX and PY becomes lower, because income is fixed and it depends on prices, i.e. A=PX*XAD+PY*YAD=B=PX*XBD+PY*YBD.

It gives PX/PY=(YAD-YBD)/(XBD-XAD)=1.572, while previously we had PX/PY=1.79.

Thus A is not able to pay as much as before for X because it will imply higher income for B, but the income is fixed. The new allocation is Pareto optimal because we cannot make B better without making worse A.

Conclusion: By fixing agents income, we demonstrate the Second Welfare Theorem, because the relationship between prices and income is eliminated. No Walras equilibrium is possible without the transfer of 13.351% A income to B, because B has worse situation than in the previous allocation. The transfer implies the shift of endowment in order to guarantee constant efficiency. If the transfer is not possible, the result can be interpreted as a Pareto optimum, but not as a Walras equilibrium.

Alternatively we can fix A and B to another original value:

```
---- VAR A          .        1.000     +INF
---- VAR B          .        1.409     +INF
```

```
A.FX = 1.409;
B.FX = 1.409;
```

The new results:

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR PX	.	1.7225	+INF	.
---- VAR PY	.	1.0955	+INF	.
---- VAR A	1.4090	1.4090	1.4090	0.1881
---- VAR B	1.4090	1.4090	1.4090	-0.1881
---- VAR XAD	.	0.3180	+INF	.
---- VAR YAD	.	0.7862	+INF	.
---- VAR XBD	.	0.6820	+INF	.
---- VAR YBD	.	0.2138	+INF	.

`Details for IRATIO=A/B=1:`

A=PX*XAD+PY*YAD =1,7225*0,318+1,0955*0,7862=1.409

B= PX*XBD +PY*YBD = 1,7225*0,682+1,0955*0,214=1.409

Conclusion: Increasing income by 41% (from 1.000 to 1.409) ⇒ nominal prices increases by 41% (from 1.223 to 1.723 for PX and from 0.777 to 1.096 for PY), but PRATIO=const and IRATIO=const ⇒ nominal value of transfer increases, but not the real one. The results in situation two (A=B=1.409) due to the money illusion may seem better than in the first situation (A=B=1). In reality, however, as the income for A and B increase, the price increases in the same speed, making both situations equally beneficial for both entities. ⇒ Utility depends on XA and YA (or on XB and YB), but not on nominal values of prices and income.

People have a tendency to view their wealth and income in nominal terms rather than in real (i.e. to recognize their real value, adjusted for inflation) terms. Economic theory calls it **money illusion**. In other words, the face value (nominal value) of money is mistaken for its purchasing power (real value). Suggesting yourself in economic decisions with nominal monetary categories (money illusion) may generate serious economic disturbances. For example: a 2% increase in nominal wages during a 4% inflation period is perceived differently (better) by employees than a 2% reduction in nominal wages in a 2% deflation period (worse), although the latter is more economically advantageous.

Exercise 4A: autarchy

(a) Set up two separate models to compute the autarchy price ratios (PRATO) for consumers A (first model) and B (second model)

```
*FOR AUTARKIA FOR CONSUMER A
A.LO = 0;
A.UP = +INF;
B.LO = 0;
B.UP = +INF;
$ONTEXT
$MODEL:AUTAR_A
$COMMODITIES:
                   PX ! AUTAR_A PRICE OF GOOD X
                   PY ! AUTAR_A PRICE OF GOOD Y
$CONSUMERS:
                    A ! CONSUMER A
$DEMAND:A        s:SIGMA_A
        E:PX     Q:XA
        E:PY     Q:YA
        D:PX     Q:THETA_A
        D:PY     Q:(1-THETA_A)
$OFFTEXT
$SYSINCLUDE mpsgeset AUTAR_A
$INCLUDE AUTAR_A.GEN
SOLVE AUTAR_A USING MCP;

PARAMETER PRICER PRICE RATIO OF THE AUTARCHY MODELS;
PRICER("AUT_A","PRATIO") = PX.L /PY.L

*FOR AUTARKIA FOR CONSUMER B
$ONTEXT
$MODEL:AUTAR_B
$COMMODITIES:
                   PX ! AUTAR_B PRICE OF GOOD X
                   PY ! AUTAR_B PRICE OF GOOD Y
$CONSUMERS:
                    B ! CONSUMER B
$DEMAND:B        s:SIGMA_B
        E:PX     Q:(1-XA)
        E:PY     Q:(1-YA)
        D:PX     Q:THETA_B
        D:PY     Q:(1-THETA_B)
$OFFTEXT
$SYSINCLUDE mpsgeset AUTAR_B
$INCLUDE AUTAR_B.GEN
SOLVE AUTAR_B USING MCP;

PARAMETER PRICER PRICE RATIO OF THE AUTARCHY MODELS;
PRICER("AUT_B","PRATIO") = PX.L /PY.L
```

solution AUTARKIA A

	LOWER	LEVEL	UPPER	MARGINAL
-- VAR PX	.	1.667	+INF	.
-- VAR PY	.	0.833	+INF	.
-- VAR A	.	1.000	+INF	.
-- VAR XAD	.	0.200	+INF	.
-- VAR YAD	.	0.800	+INF	.
-- VAR XBD	.	.	+INF	EPS
-- VAR YBD	.	.	+INF	EPS

```
PX  AUTAR_A PRICE OF GOOD X
PY  AUTAR_A PRICE OF GOOD Y
A  CONSUMER A

PARAMETER PRATIO          =        2.000  EQUILIBRIUM PRICE X I
                                          N TERMS OF Y
```

AUTARKIA B

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR PX	.	1.000	+INF	.
---- VAR PY	.	1.000	+INF	.
---- VAR B	.	1.000	+INF	.
---- VAR XAD	.	.	+INF	EPS
---- VAR YAD	.	.	+INF	EPS
---- VAR XBD	.	0.800	+INF	.
---- VAR YBD	.	0.200	+INF	.

```
PX  AUTAR_B PRICE OF GOOD X
PY  AUTAR_B PRICE OF GOOD Y
B  CONSUMER B

PARAMETER PRATIO          =        1.000  EQUILIBRIUM PRICE X I
                                          N TERMS OF Y
```

Conclusion: **(i)** Autarchy for B implies PX=PY, because this its initial allocation, i.e. agent B is not suffering from no trade. Autarchy for A implies PX=2PY, i.e. PX>PY, because the agent A demands for more X (THETA_A=0.5) than he has (XA=0.2). No trade is possible (i.e. XA=XAD), but prices reacts for the wishes of consumer. **(ii)** Fixing A and B at 1 (Second Welfare Theorem) will not change the results.

Exercise 4B:

(b) Determine parameter values in the original model where the endowment point is the equilibrium point(hint: change preferences of A to be the same as his endowment)

```
        THETA_A=0.2;

A.LO=0; A.UP=+INF;
B.LO=0; B.UP=+INF;
$INCLUDE EXCHANGE.GEN
SOLVE EXCHANGE USING MCP;
TRANSFER=100*(A.L- (PX.L*XA +PY.L*YA));
PRATIO = PX.L/PY.L;
IRATIO = A.L/B.L;
DISPLAY TRANSFER, PRATIO, IRATIO;
```

```
                    LOWER     LEVEL     UPPER    MARGINAL

--- VAR PX            .       1.000     +INF       .
--- VAR PY            .       1.000     +INF       .
--- VAR A             .       1.000     +INF       .
--- VAR B             .       1.000     +INF       .

PX  EXCHANGE PRICE OF GOOD X
PY  EXCHANGE PRICE OF GOOD Y
A   CONSUMER A
B   CONSUMER B
```

```
PARAMETER TRANSFER         =      0.000  TRANSFER GOOD X FROM
                                         CONSUMER A TO B

PARAMETER PRATIO           =      1.000  EQUILIBRIUM PRICE X I
                                         N TERMS OF Y

PARAMETER IRATIO           =      1.000  EQUILIBRIUM RATIO OF
                                         CONSUMER A INCOME TO
                                         CONSUMER B INCOME
```



Conclusion: **(i)** There is no transfer from A to B, because initial allocation is represents final preferences. In the original case, A wants more X and less Y than possess ⇒ Px↑and Py↓ **(ii)** Exercises 4a (autarchy) and 4B have similar results, i.e. demand=endowment. **(iii)** Fixing A and B at 1 (Second Welfare Theorem) will not change the results.

Exercise 4C:

*Set up a series of computations from which you can sketch the efficiency locus. Draw the Edgeworth box diagram which is consistent with these values.

```
        THETA_A=0.5;

TRANSFER=100*(A.L- (PX.L*XA +PY.L*YA));
PRATIO = PX.L/PY.L;
IRATIO = A.L/B.L;
DISPLAY TRANSFER, PRATIO, IRATIO;

*       Loop using different values of endowments of X and extract the solution
*       value of the demands:

SET             SC              SCENARIOS /SC1*SC7/;

PARAMETER       XAVALUE(SC)     VALUE SHARE OF "A" ENDOWMENT OF X
                                /SC1 0,   SC2 0.1, SC3 0.3, SC4 0.5,
                                 SC5 0.7, SC6 0.9, SC7 1/
                DEMAND(SC,*)    DEMAND BY SCENARIO;

LOOP (SC,
*       Install a XA value for the current scenario:
        XA = XAVALUE(SC);

$INCLUDE EXCHANGE.GEN
SOLVE EXCHANGE USING MCP;

*       Extract the solution value of the demands:
DEMAND(SC, "XADEM") = XAD.L;
DEMAND(SC, "YADEM") = YAD.L;
DEMAND(SC, "XBDEM") = XBD.L;
DEMAND(SC, "YBDEM") = YBD.L;
);
```

```
DISPLAY XAVALUE, DEMAND;

----    541 PARAMETER XAVALUE  VALUE SHARE OF "A" ENDOWMENT OF X

SC2 0.100,    SC3 0.300,    SC4 0.500,    SC5 0.700,    SC6 0.900,    SC7 1.000

----    541 PARAMETER DEMAND  DEMAND BY SCENARIO
```

	XADEM	YADEM	XBDEM	YBDEM
SC1	0.068	0.598	0.932	0.402
SC2	0.151	0.690	0.849	0.310
SC3	0.311	0.783	0.689	0.217
SC4	0.466	0.845	0.534	0.155
SC5	0.620	0.896	0.380	0.104
SC6	0.773	0.940	0.227	0.060
SC7	0.849	0.961	0.151	0.039

```
A.FX=1;
B.FX=1;
```

	XADEM	YADEM	XBDEM	YBDEM
SC1	0.318	0.786	0.682	0.214
SC2	0.318	0.786	0.682	0.214
SC3	0.318	0.786	0.682	0.214
SC4	0.318	0.786	0.682	0.214
SC5	0.318	0.786	0.682	0.214
SC6	0.318	0.786	0.682	0.214
SC7	0.318	0.786	0.682	0.214





Conclusion: The command LOOP allows to make a series of similar computations for different scenarios.